HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Solar Engineering



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press ■ [GTO] [•] [•] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⁑ . When you see ⁑ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press ■ [APPEND] in ALPHA mode (press ■ and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: ■ , followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [•] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: ■ [•] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing ■ and then the indirect address. Stack addresses are specified by pressing [•] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing ■ [•] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01♦LBL "SAMPLE"	■ [LBL] [ALPHA] SAMPLE [ALPHA]	01 LBL^{T}SAMPLE
02 "THIS IS A "	[ALPHA] THIS IS A [ALPHA]	02^{T}THIS IS A
03 "⊢SAMPLE"	[ALPHA] ■ [APPEND] SAMPLE	03^T ⊢ SAMPLE
04 AVIEW	■ AVIEW [ALPHA]	04 AVIEW
05 6	6	05 6
06 ENTER↑	[ENTER↑]	06 ENTER ↑
07 -2	2 [CHS]	07 -2
08 /	[÷]	08 /
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	09 ABS
10 STO IND L	[STO] ■ [•] L	10 STO IND L
11 "R3="	[ALPHA] R3= ■ [ARCL] 03	11^{T}R3=
12 ARCL 03	■ [AVIEW]	12 ARCL 03
13 AVIEW	[ALPHA]	13 AVIEW
14 RTN	■ [RTN]	14 RTN

TABLE OF CONTENTS

* Requires an additional memory module.

SOLAR-BEAM IRRADIATION

(Requires an additional memory module.)

This program enables the user to estimate solar-beam radiation impingement on a surface of any orientation and location on the earth for any day of the year. No prior knowledge of solar orbital mechanics is necessary. Solar-beam radiation rates may be estimated for any hour of the day. The program allows the user to integrate the total beam radiation over a given span of time during the day. Sunrise and sunset times may be calculated for any day of the year at any location on the earth.



Angle of incidence (θ) of beam radiation.

$$\cos\theta = \sin\delta \sin\phi \cos\delta - \sin\delta \cos\phi \sin S \cos\gamma + \cos\delta \cos\phi \cos S \cos\omega + \cos\delta \sin\phi \sin S \cos\gamma \cos\omega + \cos\delta \sin S \sin\gamma \sin\omega$$

WHERE: δ = Declination (i.e., angular position of sun at solar noon with respect to plane of equator; north is positive (see below)

ϕ = Latitude; North is positive

ω = Surface azimuth angle, the deviation of the normal to the surface from local meridian. The zero point is due south, east is positive and west is negative.

θ = Angle of incidence of beam radiation, measured between beam and normal to the plane.

Declination (δ) (Approximate)

$$\delta = 23.45 \sin[.9863\,(284 + \eta)]$$

Where: η = Numbered day of year (i.e., February 15 is 46th day of year.)

Calculation of solar angle (ω)

Solar time = Standard Time + E + 4 (L_{st} - L_{loc})

Where: E = Equation of Time

$E = 8\ \mathrm{SIN}\ (1.06\,\eta - 48) + 10\ \mathrm{SIN}\ [1.9\ (1.1\,\eta - 30)]$

L_{st} = Standard Meridian for local time zone (Standard meridians for Continental U.S. time zones are: Atlantic, 60° W; Eastern, 75° W; Central, 90° W; Mountain, 105° W; and Pacific, 120° W.)

L_{loc} = Longitude of location in question

$\omega = (12\text{-solartime}) \times 15$

Where: ω = hour angle in degrees (positive for morning and negative for afternoon.)

Zenith Angle θ_z

$$\mathrm{COS}\ \theta_z = \mathrm{SIN}\ \delta\ \mathrm{SIN}\ \phi + \mathrm{COS}\ \delta\ \mathrm{COS}\ \phi\ \mathrm{COS}\ \omega$$

Radiant Energy (G) received at surface

$$G = G_0 \times t^m\ \mathrm{COS}\ \theta$$

Where: G_0 = Solar constant 442.4 BTU HR °F FT^2

t = Transmission coefficient for unit air mass (cloudy, 0.62; mean value, 0.70; clear day, 0.81)

m = Secant of zenith angle; SEC θ_z

Time of sunrise and sunset

$$\mathrm{COS}\ \omega_s = -\mathrm{TAN}\ \Phi\ \mathrm{TAN}\ \delta$$

WHERE: ω_s = Sunrise hour angle

Sunrise solar time = $12 - \frac{\omega_s}{15}$

Sunrise standard time = Sunrise solar time - E - 4 (L_{st} - L_{loc})

Sunset solar time = $12 + \frac{\omega_s}{15}$

Sunset standard time = Sunset solar time - E - 4 (L_{st} - L_{loc})

The total irradiation during a time period

$$G_0 \int_{\omega_2}^{\omega_1} t^{\mathrm{SEC}\ \theta_z}\ \mathrm{COS}\ \theta\ d\omega = \Sigma G(\omega_1 - \omega_2)$$

EXAMPLE:



Find solar-beam radiation rate impinging on a solar collector at 10:45 a.m. and 2:20 p.m., and the total energy from 10:30 a.m. to 3:20 p.m. Also, what is the time of sunrise and sunset? The solar collector is mounted on a roof sloped 12.5° from horizontal and pointed 9° west of south. The date is September 2, 1981 and is an average clear day in Los Angeles, California. The approximate coordinates are 34° 10' north latitude and 118° 21' west longitude. The standard time meridian for Pacific Standard Time is 120° W.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] IRRAD [ALPHA]	MM.DDYYY ?
9.021981 [R/S]	LAT. ?
34.1 [R/S]	LONG.?
118.21 [R/S]	TIME MER. ?
120 [R/S]	SLOPE ?
12.5 [R/S]	AZIMUTH ?
9 [CHS] [R/S]	TRAN. COEF. ?
.7 [R/S]	A, B OR C ?
10.45 [A]	G = 267 (BTU/HR FT^2)
14.20 [A]	G = 220 (BTU/HR FT^2)
10.30 [ENTER↑]	
15.20 [ENTER↑]	
.5 [B]	ΣG = 1,122 (BTU/FT^2)

If a collector of 150 ft^2 is used, how many BTU is this.

150 [X]	181840 (BTU)
[C]	SUN R = 5:34 (AM)
[R/S]	SUN S = 18:14 (6:14 PM)

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initalize.		[XEQ] IRRAD	MM.DDYYYY ?
3	Key in the date.	date	[R/S]	LAT
4	Key in the latitude (neg. for south)			
	in degrees, minutes and seconds.	ϕ	[R/S]	LONG. ?
5.	Key in the longitude (neg. for east)			
	in degrees, minutes and seconds.	L_{loc}	[R/S]	TIME MER. ?
6	Key in the time meridian for			
	local standard time:			
	Atlantic = 60° W			
	Eastern = 75° W			
	Central = 90° W			
	Mountain = 105° W			
	Pacific = 120° W	L_{st}	[R/S]	SLOPE ?
7	Key in the slope of the plane in decimal			
	degrees.	S	[R/S]	AZIMUTH ?
8	Key in the surface azimuth in degrees,			
	minutes, seconds:			
	East is positive			
	South is zero			
	West is negative	γ	[R/S]	TRAN. COEF.?
9	Key the solar transmission coefficient:			
	Cloudy = .62			
	Mean = .70			
	Clear = .81	t	[R/S]	A, B OR C?

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
10	To find solar beam rate:			
a	Key in time of interest in hours, minutes			
	and seconds using a 24 hour clock	time	[A]	G=(BTU/FT^2)
11	To find total solar radiation over a			
	period of time:			
a	Key in starting time in hours, minutes			
	and seconds	T beg	[ENTER]	
b	Key in ending time in hours, minutes			
	and seconds	T end	[ENTER]	
c	Key in step time in decimal hours (.5			
	is good)	Δ T	[B]	ΣG=(BTU/FT^2)
12	To find sunrise and sunset:		[C]	SUN R=
			[R/S]	SUN S=

Program Listings

01♦LBL "IRRAD"	
02 SF 21	Input
03 FIX 0	
04 CF 29	
05 SF 27	
06 "MM.DDYYYY ?"	
07 CF 22	
08 RCL 06	
09 PROMPT	Date
10 FC?C 22	
11 GTO 00	
12 INT	Calculate
13 STO 01	DDY
14 LASTX	
15 FRC	
16 100	
17 *	
18 INT	
19 STO 06	
20 LASTX	
21 FRC	
22 2500	
23 *	
24 FRC	
25 STO 02	
26 RCL 01	
27 30.56	
28 *	
29 INT	
30 30	
31 -	
32 ST+ 06	
33 RCL 02	
34 ENTER↑	
35 X≠0?	
36 /	
37 1	
38 +	
39 2	
40 RCL 01	
41 X<=Y?	
42 GTO 00	
43 RCL Z	
44 ST- 06	
45♦LBL 00	
46 "LAT. ?"	
47 RCL 00	
48 PROMPT	
49 FS?C 22	Latitude

50 HR	
51 STO 00	
52 1	
53 P-R	
54 STO 11	
55 RDN	
56 STO 10	
57 "LONG. ?"	
58 RCL 02	
59 PROMPT	Longitude
60 FS?C 22	
61 HR	
62 STO 02	
63 "TIME MER. ?"	
64 RCL 01	
65 PROMPT	Time Meridian
66 STO 01	
67 "SLOPE ?"	
68 RCL 03	
69 PROMPT	Slope
70 STO 03	
71 1	
72 P-R	
73 STO 13	
74 RDN	
75 STO 12	
76 "AZIMUTH ?"	
77 RCL 04	
78 CF 22	
79 PROMPT	Azimuth
80 FS?C 22	
81 HR	
82 STO 04	
83 1	
84 P-R	
85 STO 15	
86 RDN	
87 STO 14	
88 "TRAN. COEF. ?"	
89 RCL 07	
90 PROMPT	Transmission coefficient
91 STO 07	---------------
92 RCL 06	Set up
93 284	
94 +	
95 .986	

Program Listings

96 *	
97 SIN	
98 23.45	
99 *	
100 STO 05	δ
101 1	
102 P-R	
103 STO 17	
104 RDN	
105 STO 16	
106 RCL 06	
107 81	
108 -	
109 .989	
110 *	
111 ENTER↑	
112 SIN	
113 1.5	
114 *	
115 CHS	
116 X<>Y	
117 ENTER↑	
118 COS	
119 7.53	
120 *	
121 CHS	
122 X<>Y	
123 2	
124 *	
125 SIN	
126 9.87	
127 *	
128 +	
129 +	
130 CHS	
131 60	
132 /	
133 RCL 01	
134 RCL 02	
135 -	
136 15	
137 /	
138 +	
139 STO 09	
140 "A, B OR C?"	Set up done
141 PROMPT	----------------
142♦LBL A	
143 HR	Time
144 SF 00	
145 XEQ 20	
146 RCL 19	
147 *	
148 +	
149 1/X	
150 6	
151 X<=Y?	
152 GTO 03	
153 RDN	
154 RCL 07	
155 X<>Y	
156 Y↑X	
157 442.4	Solar Constant
158 *	
159 RCL 22	
160 *	
161 X>0?	
162 GTO 04	
163♦LBL 03	
164 0	
165♦LBL 04	Output of G
166 "G="	
167 ARCL X	
168 AVIEW	
169 RTN	------------------
170♦LBL B	
171 15	$\int_{T_1}^{T_2} G$
172 *	
173 STO 25	
174 RDN	
175 HR	
176 X<>Y	
177 HR	
178 X<>Y	
179 CF 00	
180 XEQ 20	
181 STO 19	
182♦LBL 01	
183 RCL 23	
184 RCL 24	
185 -	
186 RCL 25	
187 X<=Y?	
188 GTO 02	
189 RDN	
190 STO 25	
191 SF 01	
192♦LBL 02	
193 RCL 23	
194 RCL 25	
195 2	
196 /	

Program Listings

```
197 -
198 ENTER↑
199 COS
200 RCL 21
201 *
202 X<>Y
203 1/X
204 RCL 22
205 *
206 +
207 RCL 20
208 +
209 RCL 23
210 RCL 25
211 2
212 /
213 -
214 COS
215 RCL 19
216 *
217 RCL 18
218 +
219 1/X
220 RCL 07
221 X<>Y
222 Y↑X
223 *
224 RCL 25
225 *
226 15
227 /
228 X<0?
229 0
230 ST+ 08
231 RCL 23
232 RCL 25
233 -
234 STO 23
235 FC?C 01
236 GTO 01
237 RCL 08
```

Solar Constant Output ΣG

```
238 442.4
239 *
240 "ΣG="
241 ARCL X
242 AVIEW
243 RTN
```

Common Subroutine

```
244♦LBL 20
245 RCL 09
246 +
247 12
248 X<>Y
249 -
250 15
251 *
252 STO 24
253 FS? 00
254 GTO 00
255 RDN
256 RCL 09
257 +
258 12
259 X<>Y
260 -
261 15
262 *
263 STO 23
264 0
265 STO 08
266 GTO 02
267♦LBL 00
268 1
269 P-R
270 STO 19
271 RDN
272 STO 18
273♦LBL 02
274 RCL 16
275 RCL 10
276 *
277 RCL 13
278 *
279 RCL 16
280 RCL 11
281 *
282 RCL 12
283 *
284 RCL 15
285 *
286 -
287 STO 20
288 RCL 17
289 RCL 11
290 *
291 RCL 13
292 *
293 FC? 00
294 GTO 00
295 RCL 19
296 *
297 +
298♦LBL 00
```

Program Listings

```
299 RCL 17
300 RCL 10
301 *
302 RCL 12
303 *
304 RCL 15
305 *
306 FS? 00
307 RCL 19
308 FS? 00
309 *
310 +
311 STO 21
312 RCL 17
313 RCL 12
314 *
315 RCL 14
316 *
317 FC? 00
318 GTO 00
319 RCL 18
320 *
321 +
322◆LBL 00
323 STO 22
324 RCL 16
325 RCL 10
326 *
327 STO 18
328 RCL 17
329 RCL 11
330 *
331 RTN
332◆LBL C          Sunrise/Sunset
333 RCL 00
334 TAN
335 RCL 05
336 TAN
337 *
338 CHS
339 ACOS
340 15
341 /
342 STO 24
343 12
344 RCL 09
345 -
346 STO 25
347 X<>Y
348 -
349 "SUN R="       Sunrise output
350 XEQ 00
351 RCL 25
352 RCL 24
353 +
354 "SUN S="       Sunset
355◆LBL 00         Time output
356 HMS
357 INT
358 ARCL X
359 "⊦:"
360 9
361 LASTX
362 FRC
363 100
364 *
365 INT
366 X<=Y?
367 "⊦0"
368 ARCL X
369 AVIEW
370 END
```

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ϕ	50	
	L_{st}		
	L_{loc}		
	S		
	γ		
05	δ	55	
	η		
	t		
	USED		
	USED		
10	SIN ϕ	60	
	COS ϕ		
	SIN S		
	COS S		
	SIN γ		
15	COS γ	65	
	SIN δ		
	COS δ		
	SIN ω		
	COS ω		
20	USED	70	
	USED		
	USED		
	USED		
	USED		
25	USED	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 113 USER MODE
ENG ____ FIX ____ SCI ____ ON ____ OFF ____
DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		one pt	integral
01		done	net done

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SUN ALTITUDE, AZIMUTH, SOLAR POND ABSORPTION

This program computes the Sun's azimuth and altitude (Z_n and H_c) in decimal degrees given any latitude, longitude, date and time. Then, if you wish, you can input an index of refraction for any fluid and calculate the percent of radiation which would penetrate the surface of the fluid.

The almanac equations used in this program have been checked to the end of the century for accuracy and found to be accurate to within a .2' arc.

Example:

Find the Sun's asimuth, altitude, and the fraction of the Sun's radiation which will penetrate the surface of a solar pond under the following circumstances:

Date	9/1/79
Latitude	44°34'
Longitude	123°17'
GMT	20:00:00 (Noon PST)
Index of refraction	1.33

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] ALMANAC [ALPHA]	MM.DDYYYY ?
9.011979 [R/S]	LAT ?
44.34 [R/S]	LONG ?
123.17 [R/S]	GMT ?
20 [R/S]	ZN=174.5022
[R/S]	HC=53.5985
[B]	N ?
1.33 [R/S]	%E=97.7355

Find the same information for 5 hours later.

Keystrokes:	Display:
[A]	MM.DDYYYY ?
[R/S]	LAT ?
[R/S]	LONG ?
[R/S]	GMT ?
25 [R/S]	ZN=262.9527
[R/S]	HC=18.7391
[B]	N ?
[R/S]	%E=85.1269

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load the program.			
2.	Initialize.		[XEQ] ALMANAC	MM.DDYYYY ?
3.	Key in the date.	date	[R/S]	LAT ?
4.	Key in the latitude.			
	([CHS] for south.)	L(D.MS)	[R/S]	LONG ?
5.	Key in the longitude.			
	([CHS] for east.)	λ(D.MS)	[R/S]	GMT ?
6.	Key in the Greenwich Meridian time.	GMT(H.MS)	[R/S]	ZN=
			[R/S]	HC=
7.	Press [A] and go to step 3 for another sun			
	position problem.		[A]	MM.DDYYYY ?
	OR			
8.	Press [B] for pond absorption.		[B]	N ?
9.	Key in the index of refraction n. If it			
	has been previously input, and is unchang-			
	ing, just press [R/S].	n	[R/S]	%E=
10.	For another index, go to step 8.			
	For another sun position, go to step 7.			
	Inputs that don't change can be skipped			
	by just pressing [R/S].			

Program Listings

Program	Comment
01♦LBL "ALM ANAC" 02 SF 27	Initialize
03♦LBL A 04 SF 21 05 CF 22 06 "MM.DDYY YY ?" 07 PROMPT 08 FC?C 22 09 GTO 00	Input date
10 INT 11 STO 06 12 LASTX 13 FRC 14 100 15 * 16 INT 17 STO 03 18 LASTX 19 FRC 20 1 E4 21 * 22 X<> Z 23 3056 24 % 25 INT 26 ST+ 03 27 R↑ 28 STO 04 29 RCL 06 30 3 31 X>Y? 32 1 33 RCL 04 34 4 35 / 36 FRC 37 + 38 1 39 X<>Y 40 X=Y? 41 2 42 RCL 04 43 7 44 - 45 RCL 03 46 365.25 47 / 48 + 49 STO 06 50 20 51 * 52 STO 04 53 SIN 54 4 55 * 56 50941 57 + 58 RCL 06 59 7 60 * 61 + 62 896 63 / 64 - 65 ST- 03 66 360 67 ST* 03	Calculate DOY and longitude of Moon's ascending node
68♦LBL 00 69 "LAT ?" 70 PROMPT 71 HR 72 FS?C 22 73 STO 01 74 "LONG ?" 75 PROMPT 76 HR 77 FS?C 22 78 STO 00 79 "GMT ?" 80 PROMPT 81 HR 82 15 83 * 84 FS?C 22 85 STO 02	Input other quantities
86 RCL 02 87 STO 05 88 RCL 03 89 + 90 365.25 91 / 92 118.1 93 RCL 04 94 968 95 / 96 - 97 + 98 .2 99 P-R 100 9.58	Calculate Z_n and H_c

Program Listings

```
101 -            Semidiameter
102 *
103 +
104 RCL 04
105 427
106 /
107 RCL 04
108 COS
109 -
110 8531.5
111 -
112 360
113 /
114 CHS
115 X<>Y
116 -1
117 P-R
118 RDN
119 P-R
120 R↑
121 R-P
122 RDN
123 X<>Y
124 STO 07
125 ASIN
126 STO 06      Declination
127 RDN
128 -
129 ST+ 05
130 RCL 05      GHA
131 RCL 00
132 -           LHA
133 RCL 06
134 COS
135 P-R
136 RCL 01
137 STO 06
138 X<>Y
139 P-R
140 X<> 06
141 RCL 07
142 P-R
143 X<> 06
144 +
145 ASIN
146 X<> 06
147 -
148 R-P
149 RDN
150 180
151 +
152 "ZN="
153 ARCL X
154 AVIEW
155 "HC="
156 ARCL 06
157 AVIEW
158 RTN
-----------------
159*LBL B
160 "N ?"       Input n
161 PROMPT
-----------------
162 FS?C 22
163 STO 08      Calculate %E
164 90
165 RCL 06
166 -
167 STO 05
168 SIN
169 RCL 08
170 /
171 ASIN
172 COS
173 STO 07
174 RCL 08
175 *
176 RCL 05
177 COS
178 +
179 1/X
180 X↑2
181 RCL 05
182 COS
183 RCL 08
184 *
185 RCL 07
186 +
187 1/X
188 X↑2
189 +
190 2
191 *
192 RCL 08
193 *
194 RCL 07
195 *
196 RCL 05
197 COS
198 *
199 100
200 *
201 "%E="
202 ARCL X
203 AVIEW
204 RTN
205 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Long	50	
	Lat		
	15(GMT)		
	Days		
	Ω		
05	GHA	55	
	δ		
	rlc		
	m		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 56 USER MODE

ENG ___ FIX ___ SCI ___ ON ___ OFF ___

DEG X RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ENERGY EQUIVALENTS - FUELS AND PRICES

Given an amount of fuel or energy expressed in one of the units in Table I, this program converts to an equivalent amount of another of the fuels or energy units in Table I. Also, given the price per unit of two fuels or energy units the program will convert an amount spent on one into an amount spent on the other. You may also include efficiencies between conversions. For example coal to electricity is not 100% efficient.

TABLE I

1 Barrel of Oil	= 1 BBL	= 5.8 MBTU
1000 Cubic Feet of Gas	= 1 TCF	= 1.03 MBTU
1 Gigajoule	= 1 GJ	= 1.055 MBTU
1 Short Ton of Eastern Bituminous Coal	= 1 STE	= 26 MBTU
1 Short Ton of Western Coal	= 1 STW	= 18 MBTU
1 Megawatt-hour	= 1 MWH	= 3.412 MBTU
1 Pound U_3O_8	= 1 U308	= 220 MBTU*
1 Million British Thermal Units	= 1 MBTU	

* All U^{235} atoms fissioned

Example:

How many Gigajoules can you get from 20,000 cubic feet of gas if the overall efficiency is 30%.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] ENERGY [ALPHA]	UNITS 1 ?
TCF [R/S]	$?
[R/S]	UNITS 2 ?
GJ [R/S]	$?
[R/S]	% FOR 1 TO 2
30 [R/S]	READY
20 [B]	5.86 GJ

If you wanted 10 GJ how many thousand cubic feet of gas are required?

10 [C]	31.14 TCF

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ]ENERGY	UNITS 1 ?
3	Key in the units for units 1.	Units*	[R/S]	$?
4	Key in the price per unit of unit 1.			
	Just press [R/S] if not needed.	$	[R/S]	UNITS 2 ?
5	Key in the units for units 2.	Units*	[R/S]	$?
6	Key in the price per unit of unit 2. Just			
	press [R/S] if not needed.	$	[R/S]	% FOR 1 TO 2
7	Key in the conversion efficiency to con-			
	vert from unit 1 to unit 2 if different			
	from %100, otherwise just press [R/S].	%	[R/S]	READY
8	To convert an amount from 1 to 2			
9	Key in amount of 1.	A_1	[B]	( )(UNITS 2)
10	To convert an amount from 2 to 1			
11	Key in amount of 2.	A_2	[C]	( )(UNITS 1)
12	To convert price 1 to 2			
13	Key in price 1.	P_1	[D]	$( )(UNITS 2)
14	To convert price 2 to 1			
15	Key in price 2.	P_2	[E]	$( )(UNITS 1)
16	Repeat steps 8-15 as desired.			
17	To change any or all of steps 3-6 press [A]			
	and go to step 3. For inputs that do not			
	change just press [R/S].		[A]	UNITS 1 ?
	*Key in an abbreviation from Table I.			

Program Listings

Initialize

```
 01◆LBL "ENE
RGY"
 02 SF 27
 03 1
 04 STO 02
```

Input

```
 05◆LBL A
 06 AON
 07 CF 23
 08 "UNITS 1
?"
 09 PROMPT
 10 FS?C 23
 11 ASTO 01
 12 RCL 03
 13 "$ ?"
 14 AOFF
 15 PROMPT
 16 STO 03
 17 "UNITS 2
?"
 18 AON
 19 PROMPT
 20 FS?C 23
 21 ASTO 00
 22 AOFF
 23 "$ ?"
 24 RCL 04
 25 PROMPT
 26 STO 04
 27 "% FOR 1
TO 2"
 28 CF 22
 29 PROMPT
 30 100
 31 /
 32 FS?C 22
 33 STO 02
 34 "READY"
 35 PROMPT
```

Forward

```
 36◆LBL B
 37 XEQ 01
 38 CLA
 39◆LBL 03
 40 ARCL X
 41 "⊦ "
 42 ARCL 00
 43 PROMPT
```

Backward

```
 44◆LBL C
 45 XEQ 02
 46 CLA
 47◆LBL 04
 48 ARCL X
 49 "⊦ "
 50 ARCL 01
 51 PROMPT
```

$
Forward

```
 52◆LBL D
 53 RCL 03
 54 /
 55 XEQ 01
 56 RCL 04
 57 *
 58 "$"
 59 GTO 03
```

$
Backward

```
 60◆LBL E
 61 RCL 04
 62 /
 63 XEQ 02
 64 RCL 03
 65 *
 66 "$"
 67 GTO 04
```

Conversion

```
 68◆LBL 01
 69 CF 00
 70 XEQ IND
01
 71 SF 00
 72 XEQ IND
00
 73 RCL 02
 74 *
 75 RTN
 76◆LBL 02
 77 CF 00
 78 XEQ IND
00
 79 SF 00
 80 XEQ IND
01
 81 RCL 02
 82 /
 83 RTN
```

Conversion constants

```
 84◆LBL "BBL
"
 85 5.8
 86 GTO 05
 87◆LBL "TCF
"
 88 1.03
 89 GTO 05
 90◆LBL "GJ"
 91 1.055
 92 GTO 05
```

Program Listings

```
 93♦LBL "STE
"
 94 26
 95 GTO 05
 96♦LBL "STW
"
 97 18
 98 GTO 05
 99♦LBL "MWH
"
100 3.412
101 GTO 05
102♦LBL "U3O
8"
103 220
104 GTO 05
105♦LBL "MBT
U"
106 1
107♦LBL 05
108 FS? 00
109 1/X
110 *
111 RTN
112 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	To Units	50	
	From Units		
	%		
	$ from		
	$ to		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 44 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEAT EXCHANGERS

(Requires one memory module)

This program allows analysis of counterflow, parallel flow, parallel-counterflow, and crossflow heat exchangers.

Figure 1:



Counter-Flow

Figure 2:



Parallel-Flow

Figure 3:



Parallel-Counter-Flow
(Even Number Of Tube Passes)

Figure 4:



Equations:

Heat exchanger effectiveness E is the ratio of actual heat transfer to maximum possible heat transfer.

$$E = \frac{Q}{C_{min}\,(T_{hin}-T_{cin})} = \frac{C_h\,(T_{hin}-T_{ho})}{C_{min}\,(T_{hin}-T_{cin})} = \frac{C_c\,(T_{co}-T_{cin})}{C_{min}\,(T_{hin}-T_{cin})}$$

where:

Q is the actual heat transfer.

T_{hin} and T_{cin} are the inlet temperatures of the hot and cold fluids respectively.

T_{ho} and T_{co} are the outlet temperatures of the hot and cold fluids respectively.

C_h and C_c are the heat capacities of the hot and cold fluids, respectively, e.g., $C_h = m_h \times c_{ph}$, where m_h is the flow rate and c_{ph} is the specific heat capacity of the hot fluid.

C_{min} and C_{max} (which are used later) are the smaller and larger values of C_h and C_c.

Effectiveness can be related to the product of the surface area of the heat exchanger and the overall heat transfer coefficient for specific geometries. This product is designated AU. The geometrics considered in this pac have the following correlations:

Counterflow (see figure 1)

$$E = \frac{1 - e^{-\frac{AU}{C_{min}}\left(1 - \frac{C_{min}}{C_{max}}\right)}}{1 - (C_{min}/C_{max})e^{-\frac{AU}{C_{min}}\left(1 - \frac{C_{min}}{C_{max}}\right)}}$$

For $C_{min}/C_{max} = 1$

$$E = \frac{AU/C_{min}}{1 + AU/C_{min}}$$

Parallel Flow (see figure 2)

$$E = \frac{1 - e^{-\frac{AU}{C_{min}}(1 + C_{min}/C_{max})}}{1 + C_{min}/C_{max}}$$

For $C_{min}/C_{max} = 0$, C_{min} is set to 1.

Parallel-Counterflow (well mixed with an even number of tube passes; see Figure 3)

$$E = \frac{2}{\left(1 + \frac{C_{min}}{C_{max}}\right) + \sqrt{1 + \left(\frac{C_{min}}{C_{max}}\right)^2}\left[\frac{1 + e^{-x}}{1 - e^{-x}}\right]}$$

where:

$$x = \frac{AU}{C_{min}}\sqrt{1 + \left(\frac{C_{min}}{C_{max}}\right)^2}$$

Crossflow (both fluids unmixed; see figure 4)

No exact expression exists for this case, but the following is a very good approximation. Note that an iterative solution is required for AU.

$$E = 1 - e^{\left(e^{\left(-\frac{AU}{C_{min}} \frac{C_{min}}{C_{max}} y\right)} - 1\right)\left(\frac{C_{max}}{C_{min}} \frac{1}{y}\right)}$$

where:

$$y = \left[\frac{C_{min}}{AU}\right]^{0.22}$$

References:

W.M. Kays and A.L. London, *Compact Heat Exchangers*, National Press, 1955

Eckert and Drake, *Heat and Mass Transfer*, McGraw-Hill.

Remarks:

For cases where the inlet and outlet temperatures of one of the fluids are equal(change of phase), use zero for the heat capacity of that fluid.

The solution for AU in the crossflow configuration takes significantly longer than other solutions because of the iterative technique required.

The program must be allowed to solve for all values (AU, Q, T_{co}, T_{ho}, and E). It is quite possible for the heat balance equations to yield physically meaningless solutions for a particular configuration. However, the message "2ND LAW ERR" will be displayed if the 2nd law of thermodynamics has been violated during the calculation of AU or Q.

This program is organized into five routines. The first routine performs heat balance calculations and acts as a controller for the four configuration subroutines. Each configuration subroutine has two sections that calculate AU and E for that heat exchanger. You should first load the controller, then load the configuration of interest as a separate program.

You may wish to write your own configuration routines. A routine for a configuration must be in the following format:

LBL ACON
•
•
•
(calculates AU for this configuration)
•
•
•
RTN
•
•
•
LBL ECON
•
•
•
(calculates E for this configuration)
•
•
•
END

Example:

A liquid at 168°F is to be cooled to 117°F. The liquid has a heat capacity of 0.42 Btu/LBM-°F and flows at 7700 LBM/hr. Cooling water (heat capacity = 1.00) is available at 4800 lbm/hr at 50°F. For counterflow, crossflow, parallel-counterflow, and parallel flow heat exchangers with overall coefficients of 55 Btu/hr-ft^2-°F what areas are required?

Keystrokes: (SIZE ⩾ 023)	Display:
[///] [FIX] 4	
Load main routine and counterflow subroutine.	
[XEQ] [ALPHA] HEATX [ALPHA]	TC IN=?
50 [R/S]	TH IN=?
168 [R/S]	MC=?
4800 [R/S]	MH=?
7700 [R/S]	CPC=?
1 [R/S]	CPH=?
.42 [R/S]	SELECT KEY: E AU Q TC TH
Since the temperature of the outgoing fluid is known, press the [E] key.	
[E]	THO=?
117 [R/S]	E=0.4322
[R/S]*	AU=2,198.7662
[R/S]*	Q=164,933.9999
[R/S]*	TCO=84.3612
[R/S]*	SELECT KEY: E AU Q TC TH

Keystrokes:	Display:
Since A = AU/U, calculate A.	
2198.7662 [ENTER] 55 [÷]	39.9776
Load crossflow subroutine.	
[XEQ] [ALPHA] HEATX [ALPHA]	TC IN=?
[R/S]	TH IN=?
[R/S]	MC=?
[R/S]	MH=?
[R/S]	CPC=?
[R/S]	CPH=?
[R/S]	SELECT KEY: E AU Q TC TH
[E]	THO=?
[R/S]	E=0.4322
[R/S]*	AU=2,353.6675
[R/S]*	Q=164,934.0000
[R/S]*	TCO=84.3613
[R/S]	SELECT KEY: E AU Q TC TH
2353.6675 [ENTER] 55 [÷]	42.7940

An analogus procedure will yield areas of 42.2776 ft^2 and 45.1494 ft^2 for parallel-counterflow and parallel exchanges respectively.

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program and then:		[GTO] ..	
2.	Load configuration subroutine corresponding			
	to your heat exchanger geometry		[XEQ] HEATX	TC IN=?
3.	Input inlet temperature of cold fluid	T_{cin}	[R/S]	TH IN=?
4.	Input inlet temperature of hot fluid	T_{hin}	[R/S]	MC=?
5.	Input mass flow rate of cold fluid	m_c	[R/S]	MH=?
6.	Input mass flow rate of hot fluid	m_h	[R/S]	CPC=?
7.	Input specific heat of cold fluid	C_{pc}	[R/S]	CPH=?
8.	Input specific heat of hot fluid	C_{ph}	[R/S]	SELECT KEY
				E AU Q TC TH
9.	Select the known value:			
	heat exchanger effectiveness		[A]	E=?
	area-heat transfer coefficient product		[B]	AU=?
	heat transfer		[C]	Q=?
	outlet temperature of cold fluid		[D]	TCO=?
	outlet temperature of hot fluid		[E]	THO=?
	input the known value.	E	[R/S]	
		AU	[R/S]	
		Q	[R/S]	
		TCO	[R/S]	
		THO	[R/S]	
	The four variables other than the one you			E=
	input will be output. The output order		[R/S]*	AU=
	will vary depending on which value was		[R/S]*	Q=
	input. If the 2nd law of thermodynamics		[R/S]*	TCO=
	is violated, the message "2ND LAW ERR"		[R/S]*	THO=
	will be displayed.		[R/S]*	SELECT KEY

User Instructions

SIZE: 023

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
10.	For a new problem, go to step 2 or step 9.			E AU Q TC TH
	It is not necessary to key in any values			
	which do not change. Ignore the prompts			
	and press [R/S].			
*	Press [R/S] if you do not have a printer.			

Program Listings

Heat Exchanger - Main Routine

01♦LBL "HEA TX" 02 2 03 STO 00 04 "TC IN" 05 XEQ "IN" 06 "TH IN" 07 XEQ "IN" 08 14 09 STO 00 10 "MC" 11 XEQ "IN" 12 "MH" 13 XEQ "IN" 14 "CPC" 15 XEQ "IN" 16 RCL 15 17 * 18 STO 05 19 "CPH" 20 XEQ "IN" 21 RCL 16 22 * 23 STO 06 24 "CON" 25 ASTO 22	Input values.
26♦LBL 06 27 CF 02 28 CF 03 29 CF 04 30 CF 05 31 CF 06 32 CF 21 33 SF 27 34 "SELECT KEY:" 35 AVIEW 36 SF 21 37 PSE 38♦LBL 00 39 ADV 40 "E AU Q TC TH" 41 PROMPT 42 GTO 00	Select last input.
43♦LBL A 44 SF 02 45 9 46 STO 00 47 SF 01 48 "E" 49 XEQ "IN" 50 ADV 51 GTO 16	Input E.
52♦LBL B 53 SF 03 54 10 55 STO 00 56 "AU" 57 XEQ "IN" 58 ADV 59 GTO 01	Input AU.
60♦LBL C 61 SF 04 62 11 63 STO 00 64 "Q" 65 XEQ "IN" 66 ADV 67 GTO 05	Input Q.
68♦LBL D 69 SF 05 70 12 71 STO 00 72 "TCO" 73 XEQ "IN" 74 ADV 75 GTO 14	Input TCO.
76♦LBL E 77 SF 06 78 13 79 STO 00 80 "THO" 81 XEQ "IN" 82 ADV 83 GTO 04	Input THO.
84♦LBL 16 85 FS?C 03 86 GTO 06 87 RCL 10 88 "A" 89 XEQ 08 90 STO 11 91 "AU" 92 XEQ "O"	Calculate AU.
93♦LBL 01 94 FS?C 04 95 GTO 06 96 RCL 11 97 "E" 98 XEQ 08 99 RCL 07	Calculate Q.

Program Listings

Heat Exchanger - Main Routine

```
100 *
101 RCL 04
102 RCL 03
103 -
104 *
105 STO 12
106 "Q"
107 XEQ "O"
108♦LBL 05
```

Calculate TCO.

```
109 FS?C 05
110 GTO 06
111 RCL 12
112 RCL 05
113 X≠0?
114 /
115 RCL 03
116 +
117 STO 13
118 "TCO"
119 XEQ "O"
120♦LBL 14
```

Calculate THO.

```
121 FS?C 06
122 GTO 06
123 RCL 13
124 RCL 03
125 -
126 RCL 05
127 *
128 RCL 06
129 X≠0?
130 /
131 RCL 04
132 -
133 CHS
134 STO 14
135 "THO"
136 XEQ "O"
137♦LBL 04
```

Calculate E.

```
138 FS?C 02
139 GTO 06
140 RCL 13
141 RCL 03
142 -
143 RCL 05
144 *
145 RCL 04
146 RCL 14
147 -
148 RCL 06
149 *
150 X=0?
151 X<>Y
152 RCL 04
153 RCL 03
154 -
155 /
156 RCL 05
157 RCL 06
158 X<=Y?
159 X<>Y
160 RDN
161 X=0?
162 X<> T
163 /
164 STO 10
165 SF 01
166 "E"
167 XEQ "O"
168 GTO 16
```

A0 for C_{min}=0.00.

```
169♦LBL "A0"
170 1
171 RCL 10
172 -
173 LN
174 CHS
175 RTN
```

E0 for C_{min}=0.00.

```
176♦LBL "E0"
177 1
178 RCL 11
179 CHS
180 E↑X
181 -
182 RTN
```

Find C_{min} and execute configuration subroutine.

```
183♦LBL 08
184 RCL 05
185 RCL 06
186 X>Y?
187 X<>Y
188 X<>Y
189 STO 07
190 X<>Y
191 X≠0?
192 STO 07
193 X<>Y
194 X≠0?
195 /
196 STO 09
197 SF 25
198 X=0?
199 "⊢0"
200 X≠0?
201 ARCL 22
```

Program Listings

```
202 ASTO T
203 XEQ IND
T
204 FS?C 25
205 RTN
```

Trap errors from subroutines

```
206 "2ND LAW
 ERR"
207 PROMPT
208 GTO 06
```

Input subroutine

```
209♦LBL "IN"
210 CF 22
211 1
212 ST+ 00
213 RCL IND
00
214 "⊢="
215 ASTO Y
216 "⊢?"
217 CF 21
218 AVIEW
219 SF 21
220 CLA
221 ARCL Y
222 STOP
223 STO IND
00
224 FS? 22
225 FC? 55
226 RTN
227 ARCL X
```

Print if printer is attached

```
228 PRA
229 RTN
```

Output subroutine

```
230♦LBL "O"
231 "⊢="
232 ARCL X
233 AVIEW
234 .END.
```

Program Listings

Parallel Flow Subroutine

Calculate AU.

```
01♦LBL "ACON"
02 RCL 09
03 1
04 +
05 RCL 10
06 *
07 CHS
08 1
09 +
10 LN
11 CHS
12 1
13 RCL 09
14 +
15 /
16 RCL 07
17 *
18 RTN
```

Calculate E.

```
19♦LBL "ECON"
20 1
21 +
22 RCL 11
23 RCL 07
24 /
25 *
26 CHS
27 E↑X
28 CHS
29 1
30 +
31 1
32 RCL 09
33 +
34 /
35 RTN
```

Program Listings

Counter Flow Subroutine

Calculate AU.

```
01◆LBL "ACO
N"
02 RCL 10
03 1/X
04 -
05 1
06 LASTX
07 -
08 /
09 LN
10 1
11 RCL 09
12 -
13 X=0?
14 GTO 10
15 /
16 RCL 07
17 *
18 RTN
19◆LBL 10
20 RCL 10
21 1
22 RCL 10
23 -
24 /
25 RCL 07
26 *
27 RTN
```

Calculate E.

```
28◆LBL "ECO
N"
29 1
30 -
31 RCL 11
32 RCL 07
33 /
34 *
35 E↑X
36 1
37 X<>Y
38 -
39 LASTX
40 RCL 09
41 *
42 1
43 X<>Y
44 -
45 X=0?
46 GTO 11
47 /
48 RTN
49◆LBL 11
50 RCL 11
51 RCL 07
52 /
53 ENTER↑
54 ENTER↑
55 1
56 +
57 /
58 RTN
```

Program Listings

Parallel-Counter Flow Subroutine

Calculate AU.

```
01♦LBL "ACO
N"
02 XEQ 12
03 2
04 *
05 RCL 12
06 2
07 RCL 10
08 /
09 +
10 RCL 08
11 -
12 /
13 CHS
14 1
15 +
16 LN
17 RCL 12
18 /
19 CHS
20 RCL 07
21 /
22 LASTX
23 X↑2
24 *
25 RTN
```

Calculate E.

```
26♦LBL "ECO
N"
27 XEQ 12
28 RCL 11
29 RCL 07
30 /
31 RCL 12
32 *
33 CHS
34 E↑X
35 1
36 X<>Y
37 +
38 1
39 LASTX
40 -
41 /
42 RCL 12
43 *
44 RCL 08
45 +
46 2
47 X<>Y
48 /
49 RTN
```

```
50♦LBL 12
51 RCL 09
52 1
53 +
54 STO 08
55 RCL 09
56 X↑2
57 1
58 +
59 SQRT
60 STO 12
61 RTN
```

Program Listings

Cross Flow Subroutine

Calculate AU.

```
 01♦LBL "ACO
N"
 02 0
 03 STO 19
 04 1
 05 RCL 10
 06 CHS
 07 STO 21
 08 +
 09 LN
 10 CHS
 11 STO 11
 12♦LBL 13
 13 RCL 11
 14 XEQ "ECO
N"
 15 RCL 10
 16 -
 17 STO 20
 18 RCL 19
 19 RCL 11
 20 STO 19
 21 -
 22 RCL 21
 23 RCL 20
 24 STO 21
 25 -
 26 /
 27 *
 28 ST- 11
 29 ABS
 30 1 E-4
 31 X<=Y?
 32 GTO 13
 33 RCL 11
 34 RTN
```

Calculate E.

```
 35♦LBL "ECO
N"
 36 RCL 11
 37 RCL 07
 38 /
 39 ENTER↑
 40 ENTER↑
 41 .22
 42 Y↑X
 43 RCL 09
 44 /
 45 /
 46 LASTX
 47 X<>Y
 48 CHS
 49 E↑X
 50 1
 51 -
 52 *
 53 E↑X
 54 CHS
 55 1
 56 +
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Storage Index	50	
	TC IN		
	TH IN		
05	CC	55	
	CH		
	Cmin		
	Cmax		
	Cmin/Cmax		
10	E	60	
	AU		
	Q		
	Tco		
	Tho		
15	MC	65	
	MH		
	cpc		
	cph		
	(AU_{i-1})		
20	$F(AU_i)$	70	
	$F(AU_{i-1})$		
	"CON"		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 23 TOT. REG. 102 USER MODE
ENG ___ FIX ___ SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
0		Unit conversion	
1	C		
2	C		
3	C		
4	C		
5	C		
6	C		
23		Alpha input	
55		Printer connected	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VIEW FACTOR

Given two surfaces, oriented as shown below, this program calculates the fraction of radiation leaving one surface that gets to the other, assuming a 90° angle.



The fraction of radiation that gets from 1 to 2 is the same as that which gets from 2 to 1.

Equations:

$$X = a/b,\quad Y = c/b,\quad Z = X^2 + Y^2 - 2XY\cos\Phi$$

$$\begin{aligned}
F_{A_1-A_2}(\pi Y) = & -\frac{\sin 2\Phi}{4}\left[XY\sin\Phi + \left(\frac{\pi}{2} - \Phi\right)(X^2 + Y^2)\right.\\
& + Y^2\tan^{-1}\left(\frac{X - Y\cos\Phi}{Y\sin\Phi}\right)\\
& \left. + X^2\tan^{-1}\left(\frac{Y - X\cos\Phi}{X\sin\Phi}\right)\right]\\
& + \frac{\sin^2\Phi}{4}\left\{\left(\frac{2}{\sin^2\Phi} - 1\right)\ln\left[\frac{(1 + X^2)(1 + Y^2)}{1 + Z}\right]\right.\\
& \left. + Y^2\ln\left[\frac{Y^2(1 + Z)}{(1 + Y^2)Z}\right] + X^2\ln\left[\frac{X^2(1 + X^2)^{\cos 2\Phi}}{Z(1 + Z)^{\cos 2\Phi}}\right]\right\}\\
& + Y\tan^{-1}\left(\frac{1}{Y}\right) + X\tan^{-1}\left(\frac{1}{X}\right) - \sqrt{Z}\tan^{-1}\left(\frac{1}{\sqrt{Z}}\right)\\
& + \frac{\sin\Phi\sin 2\Phi}{2}X\sqrt{1 + X^2\sin^2\Phi}\\
& \times\left[\tan^{-1}\left(\frac{X\cos\Phi}{\sqrt{1 + X^2\sin^2\Phi}}\right)\right.\\
& \left. + \tan^{-1}\left(\frac{Y - X\cos\Phi}{\sqrt{1 + X^2\sin^2\Phi}}\right)\right]\\
& + \cos\Phi\int_0^Y\sqrt{1 + \xi^2\sin^2\Phi}\left[\tan^{-1}\left(\frac{X - \xi\cos\Phi}{\sqrt{1 + \xi^2\sin^2\Phi}}\right)\right.\\
& \left. + \tan^{-1}\left(\frac{\xi\cos\Phi}{\sqrt{1 + \xi^2\sin^2\Phi}}\right)\right]d\xi
\end{aligned}$$

Example:

Find the view factor for the arrangement below:



Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 006	
[XEQ] [ALPHA] VIEW [ALPHA]	WIDTH ?
30 [R/S]	HEIGHT ?
10 [R/S]	DEPTH ?
20 [R/S]	F=0.1595

(what if the height were only 8'?)

[A]	WIDTH ?
[R/S]	HEIGHT ?
8 [R/S]	DEPTH ?
[R/S]	F=0.1379

User Instructions

SIZE: 006

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program			
2	Initialize		[XEQ] VIEW	WIDTH ?
3	Key in width	b	[R/S]	HEIGHT ?
4	Key in height	a	[R/S]	DEPTH ?
5	Key in depth	c	[R/S]	F=
6	For a new problem press [A] and go to step			
	3. Inputs that don't change can be skipped			
	by just pressing [R/S].		[A]	WIDTH ?

Program Listings

Initialize

```
01*LBL "VIE
W"
02 SF 27
```

Input

```
03*LBL A
04 CF 22
05 "WIDTH ?
"
06 PROMPT
07 FS?C 22
08 STO 00
09 "HEIGHT
?"
10 PROMPT
11 FS?C 22
12 STO 01
13 "DEPTH ?
"
14 PROMPT
15 FS?C 22
16 STO 02
```

Calculate X, Y and Z

```
17 RCL 01
18 RCL 00
19 /
20 STO 03
21 X↑2
22 RCL 02
23 RCL 00
24 /
25 STO 04
26 X↑2
27 +
28 STO 05
```

Calculate F

```
29 RCL 03
30 X↑2
31 1
32 +
33 RCL 04
34 X↑2
35 1
36 +
37 *
38 RCL 05
39 1
40 +
41 /
42 LN
43 RCL 04
44 X↑2
45 RCL 05
46 1
47 +
48 *
49 RCL 04
50 X↑2
51 1
52 +
53 /
54 RCL 05
55 /
56 LN
57 RCL 04
58 X↑2
59 *
60 +
61 RCL 03
62 X↑2
63 RCL 05
64 1
65 +
66 *
67 RCL 03
68 X↑2
69 1
70 +
71 /
72 RCL 05
73 /
74 LN
75 RCL 03
76 X↑2
77 *
78 +
79 4
80 /
81 1
82 ASIN
83 2
84 *
85 PI
86 /
87 *
88 RCL 03
89 1/X
90 ATAN
91 RCL 03
92 *
93 +
94 RCL 04
95 1/X
96 ATAN
97 RCL 04
98 *
```

Program Listings

```
 99 +
100 RCL 05
101 SQRT
102 1/X
103 ATAN
104 RCL 05
105 SQRT
106 *
107 -
108 RCL 04
109 /
110 1
111 ASIN
112 2
113 *
114 /
115 "F="          Output F
116 ARCL X
117 AVIEW
118 RTN
119 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	b	50	
	a		
	c		
	X		
	Y		
05	Z	55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 006 TOT. REG. 28 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS

This program can be used to calculate the overall heat transfer coefficient for composite tubes and walls from individual section conductances and surface coefficients.



Figure 1.—Composite tube



Figure 2.—Composite wall

Equations:

The overall heat transfer coefficient U is defined by:

$$q/L = U\ \Delta T$$

or

$$q/A = U\ \Delta T$$

where ΔT is the total temperature difference $(T_2 - T_1)$, q/L is the heat transfer per unit length of pipe, and q/A is the heat transfer per unit area of wall.

For cylinders

$$U = \frac{2\pi}{\frac{2}{h_1 D_1} + \frac{\ln(D_2/D_1)}{k_1} + \frac{\ln(D_3/D_2)}{k_2} + \ldots + \frac{\ln(D_n/D_{n-1})}{k_{n-1}} + \frac{2}{h_n D_n}}$$

For walls

$$U = \frac{1}{\frac{1}{h_1} + \frac{x_1}{k_1} + \frac{x_2}{k_2} + \ldots + \frac{x_n}{k_n} + \frac{1}{h_n}}$$

where

- h is the convective surface coefficient;
- D_n is the outside diameter of the annulus;
- k is the conductive coefficient;
- x is the thickness of a wall section.

Remarks:

These equations are for steady state heat transfer through materials with constant properties in all directions.

For composite cylinders, inputs must start with the inside convective coefficient and work out.

Zero is an invalid input for D, k, and h.

Dimensional consistency must be maintained.

Example:

A steel pipe with an inside diameter of 4 inches and a thickness of 0.5 inches has a conductivity of 25 Btu/ft-hr-°F. Two inches of asbestos (k = 0.1 Btu/hr-ft-°F) enclose the pipe bringing the total diameter to 9 inches. If the inside convective coefficient is 1000 Btu/hr-ft^2-°F and the outside coefficient is 5 Btu/hr-ft^2-°F, what is the overall heat transfer coefficient? What is the heat loss for 100 feet of pipe if ΔT is 115°F?

Keystrokes:	Display:	
[XEQ] [ALPHA] SIZE [ALPHA] 009		
[XEQ] [ALPHA] CYL [ALPHA]	NO. OF SECTS?	
2 [R/S]	D?	
4 [ENTER↑] 12 [÷] [R/S]	H?	
1000 [R/S]	D?	
5 [ENTER↑] 12 [÷] [R/S]	K?	
25 [R/S]	D?	
9 [ENTER↑] 12 [÷] [R/S]	K?	
0.1 [R/S]	H?	
5 [R/S]	U=0.98	Btu/hr-ft-°F
115 [X]	112.44	Btu/hr-ft
100 [X]	11,244.20	Btu/hr

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	For walls go to step 3, and for cylinders			
	go to step 9.			
3	Initialize for walls		[XEQ] WALLS	NO. OF SECTS?
4	Key in number of sections	N	[R/S]	H?
5	Key in the first section's			
	convective coefficient	h_1	[R/S]	X?
6	Key in thickness of the current section	x_i	[R/S]	k?
7	Key in the conductive coefficient for the			
	section of step 6	k_i	[R/S]	X? or H?
	(Repeat steps 6 and 7 for each section.			
	The prompt after the last section will be			
	"H?")			
8	Key in the last section's			
	convective coefficient	h_n	[R/S]	U=
9	Initialize for cylinders		[XEQ] CYL	NO. OF SECTS?
10	Key in number of sections	N	[R/S]	D?
11	Key in the inside section's inner			
	diameter	D_1	[R/S]	H?
12	Key in the inside convective coefficient	h_1	[R/S]	D?
13	Key in the outside diameter of the current			
	section	D_i	[R/S]	K?
14	Key in the conductive coefficient for the			
	section of step 13	k_i	[R/S]	D? or H?
	(Repeat steps 13 and 14 for each section)			
15	Key in the outside convective coefficient	h_n	[R/S]	U=

Program Listings

Initialization

```
01♦LBL "CYL
"
02 SF 00
03 GTO 00
04♦LBL "WAL
LS"
05 CF 00
06♦LBL 00
07 FIX 2
08 1
09 FS? 00
10 PI
11 STO 06
12 CLX
13 STO 08
14 FC? 00
15 GTO 01
```

Cylinders

```
16 "NO. OF
SECTS?"
17 PROMPT
18 STO 00
19 "D?"
20 PROMPT
21♦LBL A
22 STO 07
23 "H?"
24 PROMPT
25 *
26 1/X
27 ST+ 08
28 FC?C 00
29 GTO "U"
30♦LBL B
31 "D?"
32 PROMPT
33 STO 01
34 "K?"
35 PROMPT
36 1/X
37 X<>Y
38 RCL 07
39 X<>Y
40 STO 07
41 /
42 LN
43 *
44 2
45 /
46 ST- 08
47 DSE 00
48 GTO B
49 RCL 01
50 GTO A
```

Walls

```
51♦LBL 01
52 SF 00
53 "NO. OF
SECTS?"
54 PROMPT
55 STO 00
56♦LBL C
57 "H?"
58 PROMPT
59 1/X
60 ST+ 08
61 FC?C 00
62 GTO "U"
63♦LBL D
64 "X?"
65 PROMPT
66 "K?"
67 PROMPT
68 /
69 ST+ 08
70 DSE 00
71 GTO D
72 GTO C
```

Calculate U and display

```
73♦LBL "U"
74 RCL 08
75 1/X
76 RCL 06
77 *
78 STO 04
79 "U="
80 ARCL X
81 AVIEW
82 STOP
83 .END.
```

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	No. of surfaces	50	
	outside diameter		
	U		
05		55	
	1 or π		
	temp. storage		
	ΣR		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 034 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Cyl	Walls

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BLACK BODY THERMAL RADIATION

Bodies with finite temperatures emit thermal radiation. The higher the absolute temperature, the more thermal radiation emitted. Bodies which emit the maximum possible amount of energy at every wavelength for a specified temperature are said to be black bodies. While black bodies do not actually exist in nature, many surfaces may be assumed to be black for engineering considerations.



Figure 1.

Notes:

A half minute or more may be required to obtain $E_{b(0-\lambda)}$ or $E_{b(\lambda_1-\lambda_2)}$ since the integration is numerical.

Sources differ on values for constants. This could yield small discrepancies between published tables and program outputs.

Figure 1 is a representation of black body thermal emission as a function of wavelength. Note that as temperature increases, the area under the curves (total emissive power $E_{b(0-\infty)}$) increases. Also note that the wavelength of maximum emissive power λ_{max} shifts to the left as temperature increases.

This program calculates the wavelength of maximum emissive power for a given temperature, the temperature for which a given wavelength would be the wavelength of maximum emissive power, the total emissive power over all wavelengths, the emissive power at a particular wavelength, the emissive power from zero to a specified wavelength, and the emissive power between specified wavelengths.

Equations:

$$\lambda_{max} T_{\lambda_{max}} = c_3$$

$$E_{b(0-\infty)} = \sigma T^4$$

$$E_{b\lambda} = \frac{2\pi c_1}{\lambda^5 (e^{c_2/\lambda T} - 1)}$$

$$E_{b(0-\lambda)} = \int_0^{\lambda} E_{b\lambda}\, d\lambda$$

$$= 2\pi c_1 \sum_{k=1}^{\infty} -T/kc_2\, e^{-\frac{kc_2}{T\lambda}} \left[\left(\frac{1}{\lambda}\right)^3 + \frac{3T}{\lambda^2 kc_2} + \frac{6}{\lambda}\left(\frac{T}{kc_2}\right)^2 + 6\left(\frac{T}{kc_2}\right)^3 \right]$$

$$E_{b(\lambda_1-\lambda_2)} = E_{b(0-\lambda_2)} - E_{b(0-\lambda_1)}$$

where:

λ_{max} is the wavelength of maximum emissivity in microns;

T is the absolute temperature in °R or K;

$E_{b(0-\infty)}$ is the total emissive power in Btu/hr-ft^2 or Watts/cm^2;

$E_{b\lambda}$ is the emissive power at λ in Btu/hr-ft^2-μm or Watts/cm^2-μm;

$E_{b(0-\lambda)}$ is the emissive power for wavelengths less than λ in Btu/hr-ft^2 or Watts/cm^2;

$E_{b(\lambda_1-\lambda_2)}$ is the emissive power for wavelengths between λ_1 and λ_2 in Btu/hr-ft^2 or Watts/cm^2.

$c_1 = 1.8887982 \times 10^7$ Btu-μm^4/hr-ft^2
$= 5.9544 \times 10^3$ Wμm^4/cm^2

$c_2 = 2.58984 \times 10^4$ μm-°R $= 1.4388 \times 10^4$ μm-K

$c_3 = 5.216 \times 10^3$ μm-°R $= 2.8978 \times 10^3$ μm-K

$\sigma = 1.713 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.6693 \times 10^{-12}$ W/cm^2-K^4

$\sigma_{exp} = 1.731 \times 10^{-9}$ Btu/hr-ft^2-°R^4 $= 5.729 \times 10^{-12}$ W/cm^2-K^4

References: HP-67/97 Users' Library Program.

Example:

What percentage of the radiant output of a lamp is in the visible range (0.4 to 0.7 microns) if the filament of the lamp is assumed to be a black body at 2400K?

Keystrokes: (SIZE ⩾ 009)	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] BB [ALPHA]	UNITS?
SI [R/S]	TEMP?
2400 [R/S]	WAVELENGTH?
.4 [R/S]	SOLVE
[F]	WV LNTH 2?
.7 [R/S]	EbL-L=4.9679
[C]	EbTOT=188.094
[÷]	0.0264
100 [x]	2.6412

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode.		[USER]	
2.	Initialize program		[XEQ] BB	UNITS?
3.	Input code for desired units SI, or EN	SI	[R/S]	
	or EN	EN	[R/S]	TEMP?
4.	Input temperature, if temperature is	Temp	[R/S]	WAVELENGTH?
	unknown, press [R/S].			
5.	Input first wavelength, if wavelength is	λ	[R/S]	SOLVE
	unknown, press [R/S].			
6.	Calculate any or all of the following:			
	λmax for a given temperature		[A]	WLMAX=
	T such that λ is λmax for T		[B]	TEMP=
	total emissive power at T		[C]	EbTOT=
	emissive power at T and λ		[D]	EbL=
	emissive power between zero and λ		[E]	Eb0-L=
	emissive power between λ_1 and λ_2		[F]	WV LNTH 2?
		λ_2	[R/S]	EbL-L=
7.	For a new case, go to step 2.			

Program Listings

Initialize and prompt for units

```
01♦LBL "BB"
02 CLRG
03 CF 22
04 "UNITS?"
05 AON
06 PROMPT
07 AOFF
08 ASTO X
09 GTO IND
X
```

Store units

```
10♦LBL "SI"
11 5954.4
12 STO 01
13 14388
14 STO 02
15 2897.8
16 STO 03
17 5.6693 E
-12
18 STO 04
19 GTO 00
20♦LBL "EN"
21 18887982
22 STO 01
23 25998.4
24 STO 02
25 5216
26 STO 03
27 .171312
E-08
28 STO 04
```

Input prompting

```
29♦LBL 00
30 "TEMP?"
31 PROMPT
32 STO 05
33 "WAVELEN
GTH?"
34 PROMPT
35 STO 06
36 "SOLVE"
37 PROMPT
```

Calculate λmax

```
38♦LBL A
39 RCL 03
40 RCL 05
41 /
42 "WL MAX=
"
43 ARCL X
44 PROMPT
```

Calculate T(λmax)

```
45♦LBL B
46 RCL 03
47 RCL 06
48 /
49 "TEMP="
50 ARCL X
51 PROMPT
```

Calculate E_b total

```
52♦LBL C
53 RCL 05
54 X↑2
55 X↑2
56 RCL 04
57 *
58 "EbTOT="
59 ARCL X
60 PROMPT
```

Calculate $E_{b\lambda}$

```
61♦LBL D
62 RCL 01
63 ENTER↑
64 +
65 PI
66 *
67 RCL 06
68 5
69 Y↑X
70 /
71 RCL 02
72 RCL 06
73 /
74 RCL 05
75 /
76 E↑X
77 1
78 -
79 /
80 "EbL="
81 ARCL X
82 PROMPT
```

Calculate $E_b(0-\lambda)$

```
83♦LBL E
84 0
85 STO 08
86 STO 07
87♦LBL 01
88 RDN
89 CLX
90 RCL 08
91 RCL 02
92 RCL 05
93 /
94 -
95 STO 08
96 3
97 X<>Y
```

Program Listings

```
 98 /
 99 RCL 06
100 X↑2
101 /
102 LASTX
103 1/X
104 RCL 06
105 /
106 -
107 6
108 RCL 06
109 /
110 RCL 08
111 X↑2
112 /
113 -
114 6
115 RCL 08
116 X↑2
117 /
118 RCL 08
119 /
120 +
121 RCL 08
122 RCL 06
123 /
124 E↑X
125 *
126 RCL 08
127 /
128 ST+ 07
129 RCL 07
130 /
131 1 E-05
132 X<=Y?
133 GTO 01
134 RDN
135 CLX
136 RCL 07
137 ENTER↑
138 +
139 PI
140 *
141 RCL 01
142 *
143 FS?C 00
144 RTN
145 "Eb0-L="
146 ARCL X
147 PROMPT
148♦LBL F
```

Calculate $E_b(\lambda_1-\lambda_2)$

```
149 "WV LNTH
 2?"
150 PROMPT
151 ENTER↑
152 ENTER↑
153 SF 00
154 XEQ E
155 X<>Y
156 RCL 06
157 STO 00
158 RDN
159 STO 06
160 SF 00
161 XEQ E
162 -
163 ABS
164 RCL 00
165 STO 06
166 RDN
167 "EbL-L="
168 ARCL X
169 PROMPT
170 .END.
```

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	λ	50	
	C_1		
	C_2		
	C_3		
	σ		
05	T	55	
	λ,λ'		
	sum		
	kc_2/T		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 009 TOT. REG. 57 USER MODE

ENG ___ FIX 4 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	
22		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ECONOMIC BREAK EVEN FOR SOLAR EQUIPMENT

This program calculates the number of years necessary for solar equipment to pay for itself.

Equation:

$$\text{YEARS} = \frac{-\ln\left\{1 - \dfrac{\$\ \text{SPENT}\ (\%\text{INT}-\%\text{INF})}{365\,(\text{BTU/DAY})\,(\$/\text{BTU})}\right\}}{\ln\left\{1 + \dfrac{\%\text{INT}-\%\text{INF}}{1 + \%\text{INF}}\right\}}$$

where:

$ SPENT = the cost of the solar equipment.
$/BTU = the cost of purchased energy per BTU.
BTU/DAY = the amount of energy drawn from your solar equipment per day.
%INT = the current lending rate to buy equipment
%INF = the expected inflation rate for the cost of energy.
YEARS = the number of years before the solar equipment pays for itself.

Example:

Aaron B. Waters wants to buy $2000 worth of solar equipment with which he hopes to bring in 75,000 BTU per day. The cost per BTU for the energy source he is replacing is $\$3.66\text{x}10^{-6}$ $/BTU. The lending rate is 14.5% and the inflation rate is 15%. How long will it take the equipment to pay for itself?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] EBE [ALPHA]	$ SPENT ?
2000 [R/S]	$/BTU ?
3.66 [EEX] 6 [CHS] [R/S]	BTU/DAY ?
75000 [R/S]	%INT ?
14.5 [R/S]	%INF ?
15 [R/S]	21.83 YEARS

What if he spent $1500 and got 65,000 BTU/DAY?

[A]	$ SPENT ?
1500 [R/S]	$/BTU ?
[R/S]	BTU/DAY ?
65000 [R/S]	%INT ?
[R/S]	%INF ?
[R/S]	19.01 YEARS

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ] EBE	$ SPENT ?
3	Key in amount spent on solar equipment.	$ SPENT	[R/S]	$/BTU ?
4	Key in amount per BTU for the energy source			
	being replaced.	$/BTU	[R/S]	BTU/DAY ?
5	Key in the number of BTU per day to be			
	drawn from the solar equipment.	BTU/DAY	[R/S]	%INT ?
6	Key in the current lending rate.	%INT	[R/S]	%INF ?
7	Key in the expected fuel inflation rate.	%INF	[R/S]	( ) YEARS
8	For a new problem press [A] and go to step			
	3.		[A]	$ SPENT ?
	For any value which does not change just			
	press [R/S].			

Program Listings

Initialization

```
01♦LBL "EBE
"
02 SF 27
```

Input

```
03♦LBL A
04 "$ SPENT
?"
05 CF 22
06 PROMPT
07 FS?C 22
08 STO 00
09 "$/BTU ?
"
10 PROMPT
11 FS?C 22
12 STO 01
13 "BTU/DAY
?"
14 PROMPT
15 FS?C 22
16 STO 02
17 "%INT ?"
18 PROMPT
19 FS?C 22
20 STO 03
21 "%INF ?"
22 PROMPT
23 FS?C 22
24 STO 04
```

Calculate YEARS

```
25 RCL 02
26 RCL 01
27 *
28 365
29 *
30 RCL 03
31 RCL 04
32 -
33 X=0?
34 GTO 01
35 100
36 /
37 /
38 RCL 00
39 X<>Y
40 /
41 CHS
42 LN1+X
43 CHS
44 RCL 03
45 RCL 04
46 -
47 100
48 /
49 RCL 04
50 100
51 /
52 1
53 +
54 /
55 LN1+X
56 /
57♦LBL 02
58 CLA
59 ARCL X
60 "⊢ YEARS
"
61 AVIEW
62 RTN
```

Special case when %INT=%INF

```
63♦LBL 01
64 RDN
65 RCL 00
66 RCL 04
67 %
68 +
69 X<>Y
70 /
71 GTO 02
72 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	$ SPENT	50	
	$/BTU		
	BTU/DAY		
	%INT		
	%INF		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 005 TOT. REG. 25 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SOLAR PANEL ARRAY

When solar panels are installed on flat roofs or on the ground it often is necessary or desirable to arrange the collectors in several rows, one in back of another. In such an array the arrangement to prevent the south-most rows from shading the others becomes important. This program cälculates the appropriate distance between the collector arrays. Input is the Date, Latitude, Longitude, Time of Day, Local Standard Time Meridian, and the length of the solar collector panel.

Actual distance between rows, will, in final analysis, be a matter of judgement based on available space and economic conditions. For example, partial shading during the early morning and late afternoon hours in late December may be an accpetable compromise based on limited space available for panel mounting.

A most important factor in establishing the array is to establish the sun angle, S, and shade length, D_2, on an hourly and daily basis. Assuming that the array is facing south, and that you know the latitude of the location, this can be accomplished for any day of the year and time of day.

Equations:

$$N = [INT(365.25y') + INT(30.6001m') + DD + 1{,}720{,}983] - [INT(365.25(YYYY-1)) + INT(30.6001(MM+13)) + 1{,}720{,}983]$$

Where:

N=Numbered day of the year counting from Jan. 1 as day 1
MM=Month
DD=Day of the month
YYYY=Year
y'= Year-1, if MM=1 or 2
Year, if MM > 2
m'= Month+13, if MM=1 or 2
Month +1, if MM > 2

$$\delta = 23.45SIN\left[\frac{360(284+N)}{365}\right]$$

Where:

δ = Sun's declination, degrees

AST = LST + 4(LSM - LON)

Where: LON = Local Longitude

AST = Apparent Solar Time

LST = Local Standard Time

LSM = Local Standard Meridian

$$S = TAN^{-1} \frac{SIN\ \delta\ SIN\ \phi + COS\ \delta\ COS\ \phi\ COS\ w}{COS\ \delta\ SIN\ \phi\ COS\ w - SIN\ \delta\ COS\ \phi}$$

Where:

S = sun angle in a plane perpendicular to the earth and parallel to the longitude
ϕ = latitude (north positive)
w = hour angle, solar noon being zero, and each hour equaling 15° of longitude with morning positive and afternoon negative

$$V = L\ SIN\ T$$

$$D_1 = \frac{V}{TAN\ S} + L\ COS\ T$$

$$D_2 = \frac{V}{TAN\ S}$$

Where:

V = height from the horizontal to the top of solar panel, FT.

D_1 = distance from front of first row of collectors to the front of the row behind, FT.

D_2 = shade length, FT.

L = solar collector panel length, FT.



Establishing Distance Between Rows on a Flat Mounting Surface

Example:

In an array of 7' panels located at 36°25' north latitude and 97°30' west longitude with a panel tilt of 46° find V, D_1 and D_2 at 12 noon Central Standard Time on 12/21/1979.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] PANEL [ALPHA]	MM.DDYYYY?
12.211979 [R/S]	LAT ?
36.25 [R/S]	LONG ?
97.3 [R/S]	TIME ?
12 [R/S]	TIME MER ?
90 [R/S]	TILT ∠ ?
46 [R/S]	LENGTH ?
7 [R/S]	V=5.0354
[R/S]	D_1=13.6006
[R/S]	D_2=8.7380
What about at 1 PM on 6/1/1979?	
[A]	MM.DDYYYY?
6.011979 [R/S]	LAT ?
[R/S]	LONG ?
[R/S]	TIME ?
13 [R/S]	TIME MER ?
[R/S]	TILT ∠ ?
[R/S]	LENGTH ?
[R/S]	V = 5.0354
[R/S]	D_1= 6.1373
[R/S]	D_2= 1.2747

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ]PANEL	MM.DDYYYY ?
3	Key in the date.	Date	[R/S]	LAT ?
4	Key in the latitude in Degrees, Minutes			
	and Seconds (D.MS). [CHS] for south.	(D.MS)	[R/S]	LONG ?
5	Key in the longitude in D.MS. [CHS] for	D.MS)	[R/S]	TIME ?
	east. Key in the local time from a 24			
	hour clock.	t(H.MS)	[R/S]	TIME MER ?
6	Key in the time meridian:			
	60°=Atlantic Standard Time			
	75°=Eastern Standard Time			
	90°=Central Standard Time			
	105°=Mountain Standard Time			
	120°=Pacific Standard Time	(D.MS)	[R/S]	TILT∠ ?
7	Key in the angle of panel tilt.	T(D.MS)	[R/S]	LENGTH?
8	Key in the length of the panel.	L	[R/S]	V=
			[R/S]	D1=
			[R/S]	D2=
9	For a new length press [B] and go to step 8.		[B]	LENGTH?
10	To change any or all of the other variables,			
	press [A] and go to step 3.		[A]	MM.DDYYYY ?
	Skip unchanging values with [R/S].			

Program Listings

Program	Comment
01◆LBL "PANEL"	Initialization
02 SF 27	
03◆LBL A	
04 CF 22	Input
05 "MM.DDYYYY?"	
06 PROMPT	
07 FC?C 22	
08 GTO 04	
09 STO 00	Calculate DOY
10 ENTER↑	and declination
11 INT	
12 STO 07	
13 -	
14 1 E2	
15 *	
16 ENTER↑	
17 INT	
18 STO 08	
19 -	
20 1 E4	
21 *	
22 STO 09	
23 CF 02	
24◆LBL 01	
25 2	
26 RCL 07	
27 X>Y?	
28 GTO 00	
29 RCL 09	
30 1	
31 -	
32 STO 09	
33 RCL 07	
34 13	
35 +	
36 STO 07	
37 GTO 03	
38◆LBL 00	
39 RCL 07	
40 1	
41 +	
42 STO 07	
43◆LBL 03	
44 365.25	
45 RCL 09	
46 *	
47 INT	
48 30.6001	
49 RCL 07	

Program	Comment
50 *	
51 INT	
52 +	
53 RCL 08	
54 +	
55 1720982	
56 +	
57 FS? 02	
58 GTO 02	
59 STO 01	Day #
60 1	Get Day # for
61 STO 07	Jan. 1
62 STO 08	
63 SF 02	
64 GTO 01	
65◆LBL 02	
66 RCL 01	
67 1	
68 +	
69 X<>Y	
70 -	
71 STO 09	DOY
72 RCL 00	
73 CF 02	
74 360	
75 ENTER↑	
76 284	
77 RCL 09	
78 +	
79 365	
80 /	
81 *	
82 SIN	
83 23.45	
84 *	
85 STO 08	Declination
86◆LBL 04	
87 "LAT ?"	
88 PROMPT	
89 HR	
90 FS?C 22	
91 STO 05	
92 "LONG ?"	
93 PROMPT	
94 HR	
95 FS?C 22	
96 STO 02	
97 "TIME ?"	
98 PROMPT	
99 HR	
100 FS?C 22	

Program Listings

```
101 STO 03
102 "TIME ME
R ?"
103 PROMPT
104 HR
105 FS?C 22
106 STO 11
107 RCL 11
108 RCL 02
109 X<>Y
110 -
111 .0667
112 *
113 RCL 03
114 X<>Y
115 -
116 12
117 X<>Y
118 -
119 15
120 *
121 STO 06
```

hour angle

```
122 "TILT∡ ?
"
123 PROMPT
124 HR
125 FS?C 22
126 STO 04
```

```
127♦LBL B
```

Given length
calculate V
D1 and D2

```
128 "LENGTH
?"
129 CF 22
130 PROMPT
131 FS?C 22
132 STO 10
133 RCL 08
134 SIN
135 RCL 05
136 SIN
137 *
138 RCL 08
139 COS
140 RCL 05
141 COS
142 *
143 RCL 06
144 COS
145 *
146 +
147 RCL 08
148 COS
149 RCL 05
150 SIN
151 *
152 RCL 06
153 COS
154 *
155 RCL 08
156 SIN
157 RCL 05
158 COS
159 *
160 -
161 /
162 ATAN
163 STO 00
```

δ

```
164 RCL 10
165 RCL 04
166 SIN
167 *
168 STO 07
169 "V="
170 ARCL X
171 PROMPT
```

V

```
172 RCL 00
173 TAN
174 /
175 RCL 10
176 RCL 04
177 COS
178 *
179 +
180 "D1="
```

D1

```
181 ARCL X
182 PROMPT
183 RCL 07
184 RCL 00
185 TAN
186 /
187 "D2="
```

D2

```
188 ARCL X
189 PROMPT
190 .END.
```

00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Used	50	
	Used		
	Long.		
	Time		
	Tilt δ		
05	Lat.	55	
	δ		
	Used		
	AST		
	Used		
10	Length	60	
	Time Mer.		
15		65	
20		70	
		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 57 USER MODE

ENG ____ FIX 4 SCI ____ ON ____ OFF ____

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
02		Jan 1	Date

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONDUIT FLOW

This program solves for the average velocity, or the pressure drop for viscous, incompressible flow in conduits.

Equations:

$$v^2 = \frac{\Delta P/\rho}{2\left(f\,\frac{L}{D} + \frac{K_T}{4}\right)}$$

For laminar flow (Re < 2300)

$$f = 16/Re.$$

For turbulent flow (Re > 2300)

$$\frac{1}{\sqrt{f}} = 1.737 \ln \frac{D}{\varepsilon} + 2.28 - 1.737 \ln \left(4.67 \frac{D}{\varepsilon Re \sqrt{f}} + 1\right)$$

is solved by Newton's method.

$$\frac{1}{\sqrt{f_0}} = 1.737 \ln \frac{D}{\varepsilon} + 2.28$$

is used an an initial guess in the iteration.

where:

- Re is the Reynolds number, defined as $\rho Dv/\mu$;
- D is the pipe diameter;
- ε is the dimension of irregularities in the conduit surface (see table 2);
- f is the Fanning friction factor for conduit flow;
- ΔP is the pressure drop along the conduit;
- ρ is the density of the fluid;
- μ is the viscosity of the fluid;
- ν is the kinematic viscosity of the fluid and $\mu=\rho\nu$;
- L is the conduit length;
- v is the average fluid velocity;
- K_T is the total of the applicable fitting coefficients in table 1.

Table 1

Fitting Coefficients

Fitting	K
Globe valve, wide open	7.5—10
Angle valve, wide open	3.8
Gate valve, wide open	0.15—0.19
Gate valve, 3/4 open	0.85
Gate valve, 1/2 open	4.4
Gate valve, 1/4 open	20
90° elbow	0.4—0.9
Standard 45° elbow	0.35—0.42
Tee, through side outlet	1.5
Tee, straight through	.4
180° bend	1.6
Entrance to circular pipe	0.25—0.50
Sudden expansion	$(1-A_{up}/A_{dn})^2$*
Acceleration from $v=0$ to $v=v_{entrance}$	1.0

*A_{up} is the upstream area and A_{dn} is the downstream area.

Table 2

Surface Irregularities

Material	ε(feet)	ε(meters)
Drawn or Smooth Tubing	5.0×10^{-6}	1.5×10^{-6}
Commercial Steel or Wrought Iron	1.5×10^{-4}	4.6×10^{-5}
Asphalted Cast Iron	4.0×10^{-4}	1.2×10^{-4}
Galvanized Iron	5.0×10^{-4}	1.5×10^{-4}
Cast Iron	8.3×10^{-4}	2.5×10^{-4}
Wood Stave	6.0×10^{-4} to 3.0×10^{-3}	1.8×10^{-4} to 9.1×10^{-4}
Concrete	1.0×10^{-3} to 1.0×10^{-2}	3.0×10^{-4} to 3.0×10^{-3}
Riveted Steel	3.0×10^{-3} to 3.0×10^{-2}	9.1×10^{-4} to 9.1×10^{-3}

Reference:

Welty, Wicks, Wilson, *Fundamentals of Momentum, Heat and Mass Transfer*, John Wiley and Sons, Inc., 1969.

Remarks:

The correlation gives meaningless results in the region $2300 < Re < 4000$.

The solution requires an iterative procedure. The time for solution will range from 10 seconds for ΔP, to several minutes for v. The display setting is used to determine when the solution for v is adequately accurate. Time for solution of v is roughly proportional to the number or significant digits in the display setting.

If the conduit is not circular, an equivalent diameter may be calculated using the formula below:

$$D_{eq} = 4\frac{\text{cross sectional area}}{\text{wetted perimeter}}$$

Unitary consistency must be maintained.

Example:

A heat exchanger has 20, 3 meter tube passes (60 m of pipe) with 180 degrees bends connecting each pair of tubes (from table 1, $K_T = 10 \times 1.6$). The fluid is water ($\nu = 9.3 \times 10^{-7} m^2/s$, $\rho = 10^3 kg/m^3$). The surface roughness is 3×10^{-4}m and the diameter is 2.54×10^{-2}m. If the fluid velocity is 3.05 m/s, what is the pressure loss? What is the Reynolds number? What is the Fanning friction factor?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[///] [ENG] 3	
[XEQ] [ALPHA] CONDUIT [ALPHA]	U=?
9.3 [EEX] [CHS] 7 [ENTER↑]	
[EEX] 3 [X] [R/S]	RHO=?
[EEX] 3 [R/S]	E=?
3 [EEX] [CHS] 4 [R/S]	L=?
60 [R/S]	D=?
2.54 [EEX] [CHS] 2 [R/S]	KT=?
16 [R/S]	V=?
3.05 [R/S]	DP=?
[R/S]	DP=521.9E3
[R/S]	Re=83.30E3
[R/S]	F=10.18E-3

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize		[XEQ] CONDUIT	U=?
3	Key in viscosity	μ	[R/S]	RHO=?
4	Key in density	ρ	[R/S]	E=?
5	Key in surface irregularity	ε	[R/S]	L=?
6	Key in length of conduit	L	[R/S]	D=?
7	Key in equivalent diameter	D	[R/S]	KT=?
8	Key in total fitting coefficient	K_T	[R/S]	V=?
9	Key in velocity if known, otherwise			
	press [R/S]	v	[R/S]	DP=?
10	Key in pressure drop if known, otherwise			
	press [R/S]. (either v or ΔP must be			
	known)	ΔP	[R/S]	(unknown)=
11	Optional: obtain Reynolds no.		[R/S]	Re=
	Optional: obtain Fanning friction factor		[R/S]	F=
12	To change just v or ΔP		[XEQ] CHANGE	V=?
13	Go to step 9			
14	To begin a whole new problem go to step 2			

Program Listings

Input

```
 01♦LBL "CON
DUIT"
 02 "U=?"
 03 PROMPT
 04 STO 09
 05 "RHO=?"
 06 PROMPT
 07 STO 10
 08 ST/ 09
 09 "E=?"
 10 PROMPT
 11 STO 14
 12 "L=?"
 13 PROMPT
 14 STO 03
 15 "D=?"
 16 PROMPT
 17 STO 13
 18 "KT=?"
 19 PROMPT
 20 4
 21 /
 22 STO 08
 23♦LBL "CHA
NGE"
 24 CF 22
 25 "V=?"
 26 PROMPT
 27 SF 00
 28 FS? 22
 29 CF 00
 30 STO 02
 31 "DP=?"
 32 PROMPT
 33 STO 04
```

1st V

Calculate ΔP

```
 34 XEQ 09
 35 FS? 00
 36 GTO 03
 37 RCL 02
 38 X↑2
 39 *
 40 RCL 10
 41 *
 42 STO 04
 43 "DP="
 44 GTO 10
```

Iterate to find V using 1st V as guess

```
 45♦LBL 03
 46 RND
 47 STO 00
 48 XEQ 08
 49 RND
 50 RCL 00
 51 X<>Y
 52 X≠Y?
 53 GTO 03
 54 "V="
 55 RCL 02
 56 GTO 10
```

Calculate constants

```
 57♦LBL 09
 58 RCL 10
 59 RCL 13
 60 RCL 14
 61 /
 62 STO 06
 63 LN
 64 1.737
 65 STO 07
 66 *
 67 2.28
 68 +
 69 STO 12
 70 STO 05
 71 FS? 00
 72 GTO 07
```

Is flow turbulent?

```
 73♦LBL 08
 74 16
 75 RCL 02
 76 RCL 13
 77 *
 78 RCL 09
 79 /
 80 STO 01
 81 2300
 82 X<=Y?
 83 GTO 02
 84 RDN
 85 /
 86 SQRT
 87 1/X
 88 STO 05
 89 GTO 07
```

Iterate to find $\frac{1}{\sqrt{f}}$

```
 90♦LBL 02
 91 RCL 12
 92 RCL 05
 93 -
 94 4.67
 95 RCL 06
 96 *
 97 RCL 01
 98 /
 99 RCL 05
100 *
101 1
102 +
```

Program Listings


```
STO 11
LN
RCL 07
*
-
RCL 11
1/X
CHS
1
+
RCL 07
*
RCL 05
/
1
+
/
ST+ 05
RCL 05
/
ABS
 E-3
X<=Y?
GTO 02
♦LBL 07
RCL 05
1/X
X↑2
RCL 03
*
RCL 13
/
RCL 08
+
2
*
RCL 04
RCL 10
/
X<>Y
FS? 00
GTO 00
RTN
♦LBL 00
/
SQRT
STO 02
RTN
♦LBL 10
ARCL X
PROMPT
"Re="
ARCL 01
PROMPT
"F="
RCL 05
1/X
X↑2
ARCL X
PROMPT
RTN
.END.
```

Output

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	V	50	
	Re		
	V		
	L		
	ΔP		
05	$1/\sqrt{f}$	55	
	ρ/ε		
	1.737		
	$K_T/4$		
	μ		
10	ρ	60	
	used		
	$1/\sqrt{f_0}$		
	D		
	ε		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 52 USER MODE
ENG 3 FIX ___ SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		calculate V	calculate ΔP
22		calculate ΔP	calculate V

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

HEWLETT-PACKARD

HP-41C

USERS' LIBRARY SOLUTIONS

Bar Codes

Solar Engineering

SOLAR ENGINEERING

SOLAR BEAM
IRRADIATION
PROGRAM REGISTERS NEEDED: 89

HELETT PACKARD
SOLUTION BOOK:
SOLAR ENGINEERING

ROW 1 (1 : 3)

ROW 2 (4 : 6)

ROW 3 (6 : 13)

ROW 4 (14 : 22)

ROW 5 (22 : 30)

ROW 6 (30 : 41)

ROW 7 (42 : 46)

ROW 8 (46 : 57)

ROW 9 (57 : 62)

ROW 10 (63 : 64)

ROW 11 (65 : 70)

ROW 12 (71 : 76)

ROW 13 (76 : 85)

ROW 14 (86 : 88)

ROW 15 (88 : 95)

ROW 16 (95 : 102)

ROW 17 (103 : 109)

ROW 18 (110 : 119)

ROW 19 (119 : 127)

ROW 20 (128 : 138)

ROW 21 (139 : 141)

ROW 22 (142 : 149)

ROW 23 (150 : 157)

ROW 24 (158 : 166)

ROW 25 (167 : 175)

ROW 26 (176 : 183)

ROW 27 (184 : 191)

ROW 28 (192 : 201)

ROW 29 (202 : 210)

ROW 30 (211 : 221)

ROW 31 (222 : 231)

ROW 32 (231 : 238)

ROW 33 (238 : 244)

ROW 34 (244 : 253)

ROW 35 (253 : 262)

ROW 36 (263 : 272)

ROW 37 (272 : 282)

ROW 38 (283 : 293)

ROW 39 (293 : 302)

ROW 40 (303 : 311)

ROW 41 (312 : 320)

ROW 42 (321 : 329)

ROW 43 (330 : 340)

ROW 44 (341 : 349)

ROW 45 (349 : 353)

ROW 46 (354 : 359)

ROW 47 (359 : 367)

ROW 48 (367 : 370)

SUN ALTITUDE AZIMUTH
SOLAR POND ABSORPTION
PROGRAM REGISTERS NEEDED: 48

ROW 1 (1 - 2)

ROW 2 (3 - 6)

ROW 3 (6 - 12)

ROW 4 (13 - 21)

ROW 5 (22 - 29)

ROW 6 (30 - 42)

ROW 7 (43 - 50)

ROW 8 (50 - 58)

ROW 9 (59 - 66)

ROW 10 (67 - 72)

ROW 11 (73 - 78)

ROW 12 (79 - 84)

ROW 13 (85 - 92)

ROW 14 (92 - 98)

ROW 15 (99 - 106)

ROW 16 (107 - 112)

ROW 17 (113 - 124)

ROW 18 (125 - 136)

SUN ALTITUDE AZIMUTH
SOLAR POND ABSORPTION

ROW 19 (137 - 146)

ROW 20 (147 - 153)

ROW 21 (154 - 160)

ROW 22 (160 - 169)

ROW 23 (170 - 182)

ROW 24 (183 - 195)

ROW 25 (196 - 202)

ROW 26 (203 - 205)

ENERGY EQUIVALENTS
FUELS AND PRICES
PROGRAM REGISTERS NEEDED: 40

ROW 1 (1 - 3)

ROW 2 (4 - 8)

ROW 3 (8 - 14)

ROW 4 (15 - 19)

ROW 5 (20 - 27)

ROW 6 (27 - 28)

ROW 7 (28 - 34)

ROW 8 (34 - 41)

ROW 9 (41 - 48)

ROW 10 (48 - 55)

ROW 11 (55 - 63)

ROW 12 (63 - 71)

ROW 13 (71 - 79)

ROW 14 (80 - 85)

ROW 15 (85 - 88)

ROW 16 (88 - 91)

ROW 17 (91 - 94)

ROW 18 (95 - 98)

ROW 19 (99 - 101)

ROW 20 (101 - 104)

ROW 21 (104 - 108)

ROW 22 (109 - 112)

HEAT EXCHANGERS

PROGRAM REGISTERS NEEDED: 67

ROW 1 (1 - 4)

ROW 2 (4 - 6)

ROW 3 (6 - 11)

ROW 4 (11 - 14)

ROW 5 (15 - 20)

ROW 6 (20 - 26)

ROW 7 (27 - 33)

ROW 8 (33 - 34)

ROW 9 (35 - 40)

ROW 10 (40 - 44)

ROW 11 (45 - 51)

ROW 12 (51 - 57)

ROW 13 (57 - 64)

ROW 14 (64 - 70)

ROW 15 (70 - 75)

ROW 16 (76 - 81)

ROW 17 (81 - 88)

ROW 18 (88 - 94)

ROW 19 (94 - 102)

ROW 20 (103 - 110)

ROW 21 (111 - 119)

ROW 22 (119 - 129)

ROW 23 (130 - 137)

ROW 24 (138 - 148)

ROW 25 (149 - 161)

ROW 26 (162 - 168)

ROW 27 (168 - 175)

ROW 28 (176 - 183)

ROW 29 (184 - 196)

ROW 30 (197 - 203)

ROW 31 (204 - 206)

ROW 32 (206 - 210)

ROW 33 (211 - 216)

ROW 34 (217 - 224)

ROW 35 (225 - 230)

ROW 36 (231 - 234)

HEAT EXCHANGERS
COUNTER FLOW
PROGRAM REGISTERS NEEDED: 8

ROW 1 (1 - 6)

ROW 2 (7 - 19)

ROW 3 (19 - 25)

ROW 4 (26 - 36)

HEAT EXCHANGERS
CROSS FLOW
PROGRAM REGISTERS NEEDED: 14

ROW 1 (1 – 5)

ROW 2 (6 – 14)

ROW 3 (14 – 22)

ROW 4 (22 – 30)

ROW 5 (30 – 35)

ROW 6 (35 – 44)

ROW 7 (45 – 57)

ROW 8 (57 – 57)

HEAT EXCHANGERS
PAR–COUNTER FLOW
PROGRAM REGISTERS NEEDED: 12

ROW 1 (1 – 4)

ROW 2 (5 – 17)

ROW 3 (18 – 26)

ROW 4 (26 – 34)

ROW 5 (35 – 47)

ROW 6 (48 – 60)

ROW 7 (61 – 62)

HEAT EXCHANGERS
PARALLEL FLOW
PROGRAM REGISTERS NEEDED: 8

ROW 1 (1 - 6)

ROW 2 (7 - 19)

ROW 3 (19 - 25)

ROW 4 (26 - 36)

VIEW FACTOR

PROGRAM REGISTERS NEEDED: 23

ROW 1 (1 - 4)

ROW 2 (4 - 8)

ROW 3 (9 - 12)

ROW 4 (13 - 17)

ROW 5 (18 - 30)

ROW 6 (31 - 43)

ROW 7 (44 - 56)

ROW 8 (57 - 69)

ROW 9 (70 - 82)

ROW 10 (83 - 95)

ROW 11 (96 - 108)

ROW 12 (109 - 118)

ROW 13 (119 - 119)

HEAT TRANSFER THROUGH COMPOSITE CYLINDERS AND WALLS

PROGRAM REGISTERS NEEDED: 26

ROW 1 (1 – 4)

ROW 2 (4 – 8)

ROW 3 (9 – 16)

ROW 4 (16 – 18)

ROW 5 (19 – 26)

ROW 6 (27 – 32)

ROW 7 (33 – 43)

ROW 8 (44 – 50)

ROW 9 (51 – 53)

ROW 10 (53 – 59)

ROW 11 (60 – 65)

ROW 12 (66 – 72)

ROW 13 (72 – 79)

RCW 14 (79 – 83)

BLACK BODY THERMAL RADIATION

PROGRAM REGISTERS NEEDED: 48

ROW 1 (1 - 4)

ROW 2 (4 - 10)

ROW 3 (10 - 13)

ROW 4 (13 - 17)

ROW 5 (17 - 20)

ROW 6 (20 - 23)

ROW 7 (23 - 27)

ROW 8 (27 - 30)

ROW 9 (30 - 33)

ROW 10 (33 - 36)

ROW 11 (36 - 42)

ROW 12 (42 - 49)

ROW 13 (49 - 57)

ROW 14 (58 - 62)

ROW 15 (63 - 75)

ROW 16 (76 - 83)

ROW 17 (83 - 95)

ROW 18 (96 - 108)

ROW 19 (109 - 121)

ROW 20 (122 - 131)

ROW 21 (131 - 141)

ROW 22 (142 - 146)

ROW 23 (147 - 149)

ROW 24 (149 - 158)

ROW 25 (159 - 167)

ROW 26 (167 - 170)

ECONOMIC BREAK EVEN
FOR SOLAR EQUIPMENT
PROGRAM REGISTERS NEEDED: 21

ROW 1 (1 - 4)

ROW 2 (4 - 7)

ROW 3 (8 - 12)

ROW 4 (13 - 15)

ROW 5 (16 - 21)

ROW 6 (21 - 27)

ROW 7 (28 - 35)

ROW 8 (36 - 47)

ROW 9 (47 - 57)

ROW 10 (58 - 62)

ROW 11 (63 - 72)

SOLAR PANEL ARRAY

PROGRAM REGISTERS NEEDED: 46

ROW 1 (1 - 3)

ROW 2 (4 - 5)

ROW 3 (6 - 14)

ROW 4 (15 - 24)

ROW 5 (25 - 35)

ROW 6 (36 - 44)

ROW 7 (44 - 49)

ROW 8 (50 - 56)

ROW 9 (57 - 65)

ROW 10 (66 - 75)

ROW 11 (76 - 83)

ROW 12 (83 - 88)

ROW 13 (89 - 94)

ROW 14 (95 - 100)

ROW 15 (100 - 102)

ROW 16 (103 - 111)

ROW 17 (111 - 121)

ROW 18 (122 - 126)

ROW 19 (127 – 129)

ROW 20 (130 – 141)

ROW 21 (142 – 154)

ROW 22 (155 – 167)

ROW 23 (168 – 177)

ROW 24 (178 – 186)

ROW 25 (187 – 190)

CONDUIT FLOW

PROGRAM REGISTERS NEEDED: 39

ROW 1 (1 - 2)

ROW 2 (2 - 8)

ROW 3 (8 - 14)

ROW 4 (15 - 20)

ROW 5 (21 - 24)

ROW 6 (24 - 30)

ROW 7 (31 - 36)

ROW 8 (36 - 44)

ROW 9 (45 - 54)

ROW 10 (54 - 63)

ROW 11 (64 - 69)

ROW 12 (70 - 79)

ROW 13 (80 - 88)

ROW 14 (89 - 97)

ROW 15 (98 - 110)

ROW 16 (111 - 122)

ROW 17 (123 - 131)

ROW 18 (132 - 143)

ROW 19 (144 – 154)

ROW 20 (154 – 161)

ROW 21 (161 – 165)

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

HEWLETT PACKARD

Rev. B

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41C programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41C, HP has an extensive library of "Application Pacs". These programs transform your HP-41C into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, minature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41C.

You can choose from:

Aviation
Clinical Lab
Circuit Analysis
Financial Decisions
Mathematics
Structural Analysis
Surveying
Securities
Statistics
Stress Analysis
Games
Home Management
Machine Design
Navigation
Real Estate
Thermal and Transport Science

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41C Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*** Users' Library Solutions Books**

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales
Home Construction Estimating
Lending, Saving and Leasing
Real Estate
Small Business
Geometry
High-Level Math
Test Statistics
Antennas
Chemical Engineering
Control Systems
Electrical Engineering
Fluid Dynamics and Hydraulics
Civil Engineering
Heating, Ventilating & Air Conditioning
Mechanical Engineering
Solar Engineering
Calendars
Cardiac/Pulmonary
Chemistry
Games
Optometry I (General)
Optometry II (Contact Lens)
Physics
Surveying

* Some books require additional memory modules to accomodate all programs.

SOLAR ENGINEERING

HEWLETT
PACKARD

00041-90138 Rev. B 5/81 Printed in U.S.A.